

# The Kuratowski Closure-Complement Theorem

Peter Gammie and Gianpaolo Gioiosa

June 16, 2019

## Contents

|          |  |           |
|----------|--|-----------|
| <b>1</b> | <b>Introduction</b>  | <b>1</b>  |
| <b>2</b> | <b>Interiors and unions</b>                                    | <b>1</b>  |
| <b>3</b> | <b>Additional facts about the rationals and reals</b>          | <b>3</b>  |
| <b>4</b> | <b>Kuratowski's result</b>                                     | <b>3</b>  |
| <b>5</b> | <b>A corollary of Kuratowski's result</b>                      | <b>7</b>  |
| <b>6</b> | <b>Chagrov's result</b>  | <b>8</b>  |
| 6.1      | Discrete spaces . . . . .                                      | 10        |
| 6.2      | Partition spaces . . . . .                                     | 10        |
| 6.3      | Extremally disconnected and open unresolvable spaces . . . . . | 12        |
| 6.4      | Extremally disconnected spaces . . . . .                       | 15        |
| 6.5      | Open unresolvable spaces . . . . .                             | 17        |
| 6.6      | Kuratowski spaces . . . . .                                    | 18        |
| 6.7      | Chagrov's theorem . . . . .                                    | 20        |
|          | <b>References</b>  | <b>20</b> |

## 1 Introduction

We discuss a topological curiosity discovered by [Kuratowski \(1922\)](#): the fact that the number of distinct operators on a topological space generated by compositions of closure and complement never exceeds 14, and is exactly 14 in the case of  $\mathbb{R}$ . In addition, we prove a theorem due to [Chagrov \(1982\)](#) that classifies topological spaces according to the number of such operators they support.

Kuratowski's result, which is exposted in [Whitty \(2015\)](#) and Chapter 7 of [Chamberland \(2015\)](#), has already been treated in Mizar — see [Bagińska and Grabowski \(2003\)](#) and [Grabowski \(2004\)](#). To the best of our knowledge, we are the first to mechanize Chagrov's result.

Our work is based on a presentation of Kuratowski's and Chagrov's results by [Gardner and Jackson \(2008\)](#).

We begin with some preliminary facts pertaining to the relationship between interiors of unions and unions of interiors (§2) and the relationship between  $\mathbb{Q}$  and  $\mathbb{R}$  (§3). We then prove Kuratowski's result (§4) and the corollary that at most 7 distinct operators on a topological space can be generated by compositions of closure and interior (§5). Finally, we prove Chagrov's result (§6).

## 2 Interiors and unions

### definition

*boundary* :: 'a::topological\_space set  $\Rightarrow$  'a set

### where

*boundary*  $X = \text{closure } X - \text{interior } X$

**lemma** *boundary\_empty*:  
**shows** *boundary* {} = {}  
**unfolding** *boundary\_def* **by** *simp*

**definition**

*exterior* :: 'a::topological\_space set  $\Rightarrow$  'a set

**where**

*exterior* X = - (interior X  $\cup$  boundary X)

**lemma** *interior\_union\_boundary*:

**shows** interior (X  $\cup$  Y) = interior X  $\cup$  interior Y

$\longleftrightarrow$  boundary X  $\cap$  boundary Y  $\subseteq$  boundary (X  $\cup$  Y) (**is** (?lhs1 = ?lhs2)  $\longleftrightarrow$  ?rhs)

**proof**(rule iffI[OF - subset\_antisym[OF subsetI]])

**assume** ?lhs1 = ?lhs2 **then show** ?rhs **by** (force simp: boundary\_def)

**next**

**fix** x

**assume** ?rhs **and** x  $\in$  ?lhs1

**have** x  $\in$  ?lhs2 **if** x  $\notin$  interior X x  $\notin$  interior Y

**proof**(cases x  $\in$  boundary X  $\cap$  boundary Y)

**case** True **with** ⟨?rhs⟩ ⟨x  $\in$  ?lhs1⟩ **show** ?thesis **by** (simp add: boundary\_def subset\_iff)

**next**

**case** False **then consider** (X) x  $\notin$  boundary X | (Y) x  $\notin$  boundary Y **by** blast

**then show** ?thesis

**proof** cases

**case** X

**from** X ⟨x  $\notin$  interior X⟩ **have** x  $\in$  exterior X **by** (simp add: exterior\_def)

**from** ⟨x  $\notin$  boundary X⟩ ⟨x  $\in$  exterior X⟩ ⟨x  $\notin$  interior X⟩

**obtain** U **where** open U U  $\subseteq$  - X x  $\in$  U

**by** (metis ComplI DiffI boundary\_def closure\_interior interior\_subset open\_interior)

**from** ⟨x  $\in$  interior (X  $\cup$  Y)⟩ **obtain** U' **where** open U' U'  $\subseteq$  X  $\cup$  Y x  $\in$  U'

**by** (meson interiorE)

**from** ⟨U  $\subseteq$  - X⟩ ⟨U'  $\subseteq$  X  $\cup$  Y⟩ **have** U  $\cap$  U'  $\subseteq$  Y **by** blast

**with** ⟨x  $\notin$  interior Y⟩ ⟨open U'⟩ ⟨open U⟩ ⟨x  $\in$  U'⟩ ⟨x  $\in$  U⟩ **show** ?thesis

**by** (meson IntI interiorI open\_Int)

**next**

**case** Y

**from** Y ⟨x  $\notin$  interior Y⟩ **have** x  $\in$  exterior Y **by** (simp add: exterior\_def)

**from** ⟨x  $\notin$  boundary Y⟩ ⟨x  $\in$  exterior Y⟩ ⟨x  $\notin$  interior Y⟩

**obtain** U **where** open U U  $\subseteq$  - Y x  $\in$  U

**by** (metis ComplI DiffI boundary\_def closure\_interior interior\_subset open\_interior)

**from** ⟨x  $\in$  interior (X  $\cup$  Y)⟩ **obtain** U' **where** open U' U'  $\subseteq$  X  $\cup$  Y x  $\in$  U'

**by** (meson interiorE)

**from** ⟨U  $\subseteq$  - Y⟩ ⟨U'  $\subseteq$  X  $\cup$  Y⟩ **have** U  $\cap$  U'  $\subseteq$  X **by** blast

**with** ⟨x  $\notin$  interior X⟩ ⟨open U'⟩ ⟨open U⟩ ⟨x  $\in$  U'⟩ ⟨x  $\in$  U⟩ **show** ?thesis

**by** (meson IntI interiorI open\_Int)

**qed**

**qed**

**with** ⟨x  $\in$  ?lhs1⟩ **show** x  $\in$  ?lhs2 **by** blast

**next**

**show** ?lhs2  $\subseteq$  ?lhs1 **by** (simp add: interior\_mono)

**qed**

**lemma** *interior\_union\_closed\_intervals*:

**fixes** a :: 'a::ordered\_euclidean\_space

**assumes** b < c

**shows** interior ({a..b}  $\cup$  {c..d}) = interior {a..b}  $\cup$  interior {c..d}

**using** *assms* **by** (subst interior\_union\_boundary; auto simp: boundary\_def)

### 3 Additional facts about the rationals and reals

**lemma** *Rat\_real\_limpt*:

**fixes**  $x :: \text{real}$

**shows**  $x \text{ islimpt } \mathbb{Q}$

**proof**(*rule islimptI*)

**fix**  $T$  **assume**  $x \in T$  **open**  $T$

**then obtain**  $e$  **where**  $0 < e$  **and**  $\text{ball}: |x' - x| < e \longrightarrow x' \in T$  **for**  $x'$  **by** (*auto simp: open\_real*)

**from**  $(0 < e)$  **obtain**  $q$  **where**  $x < \text{real.of\_rat } q \wedge \text{real.of\_rat } q < x + e$  **using** *of\\_rat\\_dense* **by force**

**with**  $\text{ball}$  **show**  $\exists y \in \mathbb{Q}. y \in T \wedge y \neq x$  **by force**

**qed**

**lemma** *Rat\_closure*:

**shows**  $\text{closure } \mathbb{Q} = (\text{UNIV} :: \text{real set})$

**unfolding** *closure\_def* **using** *Rat\_real\_limpt* **by blast**

**lemma** *Rat\_interval\_closure*:

**fixes**  $x :: \text{real}$

**assumes**  $x < y$

**shows**  $\text{closure } (\{x <..<y\} \cap \mathbb{Q}) = \{x..y\}$

**using** *assms*

**by** (*metis (no\_types, lifting) Rat\_closure closure\_closure closure\_greaterThanLessThan closure\_mono inf\_le1 inf\_top.right\_ne open\_Int\_closure\_subset open\_real\_greaterThanLessThan subset\_antisym*)

**lemma** *Rat\_not\_open*:

**fixes**  $T :: \text{real set}$

**assumes** *open*  $T$

**assumes**  $T \neq \{\}$

**shows**  $\neg T \subseteq \mathbb{Q}$

**using** *assms* **by** (*simp add: countable\_rat open\_minus\_countable subset\_eq*)

**lemma** *Irrat\_dense\_in\_real*:

**fixes**  $x :: \text{real}$

**assumes**  $x < y$

**shows**  $\exists r \in -\mathbb{Q}. x < r \wedge r < y$

**using** *assms* *Rat\_not\_open* [**where**  $T = \{x <..<y\}$ ] **by force**

**lemma** *closed\_interval\_Int\_compl*:

**fixes**  $x :: \text{real}$

**assumes**  $x < y$

**assumes**  $y < z$

**shows**  $-\{x..y\} \cap -\{y..z\} = -\{x..z\}$

**using** *assms* **by auto**

### 4 Kuratowski's result

We prove that at most 14 distinct operators can be generated by compositions of *closure* and complement. For convenience, we give these operators short names and try to avoid pointwise reasoning. We treat the *interior* operator at the same time.

**declare** *o\_apply* [*simp del*]

**definition**  $C :: 'a :: \text{topological\_space set} \Rightarrow 'a \text{ set}$  **where**  $C X = - X$

**definition**  $K :: 'a :: \text{topological\_space set} \Rightarrow 'a \text{ set}$  **where**  $K X = \text{closure } X$

**definition**  $I :: 'a :: \text{topological\_space set} \Rightarrow 'a \text{ set}$  **where**  $I X = \text{interior } X$

**lemma** *C\_C*:

**shows**  $C \circ C = id$

**by** (*simp add: fun\_eq\_iff C\_def o\_apply*)

**lemma** *K\_K*:

**shows**  $K \circ K = K$

**by** (*simp add: fun\_eq\_iff K\_def o\_apply*)

**lemma** *I\_I*:

**shows**  $I \circ I = I$

**unfolding** *I\_def* **by** (*simp add: o\_def*)

**lemma** *I\_K*:

**shows**  $I = C \circ K \circ C$

**unfolding** *C\_def I\_def K\_def* **by** (*simp add: o\_def interior\_closure*)

**lemma** *K\_I*:

**shows**  $K = C \circ I \circ C$

**unfolding** *C\_def I\_def K\_def* **by** (*simp add: o\_def interior\_closure*)

**lemma** *K\_I\_K\_I*:

**shows**  $K \circ I \circ K \circ I = K \circ I$

**unfolding** *C\_def I\_def K\_def*

**by** (*clarsimp simp: fun\_eq\_iff o\_apply closure\_minimal closure\_mono closure\_subset interior\_maximal interior\_subset subset\_antisym*)

**lemma** *I\_K\_I\_K*:

**shows**  $I \circ K \circ I \circ K = I \circ K$

**unfolding** *C\_def I\_def K\_def*

**by** (*simp add: fun\_eq\_iff o\_apply*)

(*metis (no\_types) closure\_closure closure\_mono closure\_subset interior\_maximal interior\_mono interior\_subset open\_interior subset\_antisym*)

**lemma** *K\_mono*:

**assumes**  $x \subseteq y$

**shows**  $K x \subseteq K y$

**using** *assms* **unfolding** *K\_def* **by** (*simp add: closure\_mono*)

The following lemma embodies the crucial observation about compositions of  $C$  and  $K$ :

**lemma** *KCKCKCK\_KCK*:

**shows**  $K \circ C \circ K \circ C \circ K \circ C \circ K = K \circ C \circ K$  (**is** *?lhs = ?rhs*)

**proof**(*rule ext[OF equalityI]*)

**fix**  $x$

**have**  $(C \circ K \circ C \circ K \circ C \circ K) x \subseteq ?rhs x$  **by** (*simp add: C\_def K\_def closure\_def o\_apply*)

**then have**  $(K \circ (C \circ K \circ C \circ K \circ C \circ K)) x \subseteq (K \circ ?rhs) x$  **by** (*simp add: K\_mono o\_apply*)

**then show**  $?lhs x \subseteq ?rhs x$  **by** (*simp add: K\_K o\_assoc*)

**next**

**fix**  $x :: 'a::topological\_space\ set$

**have**  $(C \circ K \circ C \circ K) x \subseteq K x$  **by** (*simp add: C\_def K\_def closure\_def o\_apply*)

**then have**  $(K \circ (C \circ K \circ C \circ K)) x \subseteq (K \circ K) x$  **by** (*simp add: K\_mono o\_apply*)

**then have**  $(C \circ (K \circ K)) x \subseteq (C \circ (K \circ (C \circ K \circ C \circ K))) x$  **by** (*simp add: C\_def o\_apply*)

**then have**  $(K \circ (C \circ (K \circ K))) x \subseteq (K \circ (C \circ (K \circ (C \circ K \circ C \circ K)))) x$  **by** (*simp add: K\_mono o\_apply*)

**then show**  $?rhs x \subseteq ?lhs x$  **by** (*simp add: K\_K o\_assoc*)

**qed**

The inductive set  $CK$  captures all operators that can be generated by compositions of  $C$  and  $K$ . We shallowly embed the operators; that is, we identify operators up to extensional equality.

**inductive** *CK* :: (*'a::topological\_space set*  $\Rightarrow$  *'a set*)  $\Rightarrow$  *bool* **where**

```

  CK C
| CK K
| [ CK f; CK g ]  $\implies$  CK (f  $\circ$  g)

```

**declare** *CK.intros*[intro!]

**lemma** *CK\_id*[intro!]:

```

  CK id
by (metis CK.intros(1) CK.intros(3) C_C)

```

The inductive set *CK\_nf* captures the normal forms for the 14 distinct operators.

**inductive** *CK\_nf* :: ('a::topological\_space set  $\Rightarrow$  'a set)  $\Rightarrow$  bool **where**

```

  CK_nf id
| CK_nf C
| CK_nf K
| CK_nf (C  $\circ$  K)
| CK_nf (K  $\circ$  C)
| CK_nf (C  $\circ$  K  $\circ$  C)
| CK_nf (K  $\circ$  C  $\circ$  K)
| CK_nf (C  $\circ$  K  $\circ$  C  $\circ$  K)
| CK_nf (K  $\circ$  C  $\circ$  K  $\circ$  C)
| CK_nf (C  $\circ$  K  $\circ$  C  $\circ$  K  $\circ$  C)
| CK_nf (K  $\circ$  C  $\circ$  K  $\circ$  C  $\circ$  K)
| CK_nf (C  $\circ$  K  $\circ$  C  $\circ$  K  $\circ$  C  $\circ$  K)
| CK_nf (K  $\circ$  C  $\circ$  K  $\circ$  C  $\circ$  K  $\circ$  C)
| CK_nf (C  $\circ$  K  $\circ$  C  $\circ$  K  $\circ$  C  $\circ$  K  $\circ$  C)

```

**declare** *CK\_nf.intros*[intro!]

**lemma** *CK\_nf\_set*:

```

  shows {f . CK_nf f} = {id, C, K, C  $\circ$  K, K  $\circ$  C, C  $\circ$  K  $\circ$  C, K  $\circ$  C  $\circ$  K, C  $\circ$  K  $\circ$  C  $\circ$  K, K  $\circ$  C  $\circ$  K  $\circ$  C, C  $\circ$  K  $\circ$  C  $\circ$  K  $\circ$  C, K  $\circ$  C  $\circ$  K  $\circ$  C  $\circ$  K, C  $\circ$  K  $\circ$  C  $\circ$  K  $\circ$  C  $\circ$  K, K  $\circ$  C  $\circ$  K  $\circ$  C  $\circ$  K  $\circ$  C  $\circ$  K, C  $\circ$  K  $\circ$  C  $\circ$  K  $\circ$  C  $\circ$  K  $\circ$  C}

```

**by** (auto simp: *CK\_nf.simps*)

That each operator generated by compositions of *C* and *K* is extensionally equivalent to one of the normal forms captured by *CK\_nf* is demonstrated by means of an induction over the construction of *CK\_nf* and an appeal to the facts proved above.

**theorem** *CK\_nf*:

```

  CK f  $\longleftrightarrow$  CK_nf f

```

**proof**(rule *iffI*)

**assume** *CK f* **then show** *CK\_nf f*

**by** *induct*

```

  (elim CK_nf.cases; clarsimp simp: id_def[symmetric] C_C K_K KCKCKCK_KCK o_assoc; simp add:
o_assoc[symmetric]; clarsimp simp: C_C K_K KCKCKCK_KCK o_assoc
| blast)+

```

**next**

**assume** *CK\_nf f* **then show** *CK f* **by** *induct* (auto simp: *id\_def[symmetric]*)

**qed**

**theorem** *CK\_card*:

```

  shows card {f. CK f}  $\leq$  14

```

**by** (auto simp: *CK\_nf CK\_nf\_set card\_insert intro!: le\_trans[OF card\_Diff1\_le]*)

We show, using the following subset of  $\mathbb{R}$  (an example taken from [Rusin \(2001\)](#)) as a witness, that there exist topological spaces on which all 14 operators are distinct.

**definition**

```

  RRR :: real set

```

where

$$RRR = \{0 < .. < 1\} \cup \{1 < .. < 2\} \cup \{3\} \cup (\{5 < .. < 7\} \cap \mathbb{Q})$$

The following facts allow the required proofs to proceed by *simp*:

**lemma** *RRR\_closure*:

**shows** *closure*  $RRR = \{0..2\} \cup \{3\} \cup \{5..7\}$

**unfolding** *RRR\_def* **by** (*force simp: closure\_insert Rat\_interval\_closure*)

**lemma** *RRR\_interior*:

*interior*  $RRR = \{0 < .. < 1\} \cup \{1 < .. < 2\}$  (**is** *?lhs = ?rhs*)

**proof**(*rule equalityI[OF subsetI subsetI]*)

**fix** *x* **assume**  $x \in ?lhs$

**then obtain** *T* **where** *open T* **and**  $x \in T$  **and**  $T \subseteq RRR$  **by** (*blast elim: interiorE*)

**then obtain** *e* **where**  $0 < e$  **and**  $\text{ball } x \ e \subseteq T$  **by** (*blast elim!: openE*)

**from**  $\langle x \in T \rangle \langle 0 < e \rangle \langle \text{ball } x \ e \subseteq T \rangle \langle T \subseteq RRR \rangle$

**have** *False* **if**  $x = 3$

**using** *that* **unfolding** *RRR\_def ball\_def*

**by** (*auto dest!: subsetD[where c=min (3 + e/2) 4] simp: dist\_real\_def*)

**moreover**

**from** *Irrat\_dense\_in\_real[where x=x and y=x + e/2] <0 < e>*

**obtain** *r* **where**  $r \in -\mathbb{Q} \wedge x < r \wedge r < x + e / 2$  **by** *auto*

**with**  $\langle x \in T \rangle \langle \text{ball } x \ e \subseteq T \rangle \langle T \subseteq RRR \rangle$

**have** *False* **if**  $x \in \{5 < .. < 7\} \cap \mathbb{Q}$

**using** *that* **unfolding** *RRR\_def ball\_def*

**by** (*force simp: dist\_real\_def dest: subsetD[where c=r]*)

**moreover note**  $\langle x \in \text{interior } RRR \rangle$

**ultimately show**  $x \in ?rhs$

**unfolding** *RRR\_def* **by** (*auto dest: subsetD[OF interior\_subset]*)

**next**

**fix** *x* **assume**  $x \in ?rhs$

**then show**  $x \in ?lhs$

**unfolding** *RRR\_def interior\_def* **by** (*auto intro: open\_real\_greaterThanLessThan*)

**qed**

**lemma** *RRR\_interior\_closure[simplified]*:

**shows** *interior*  $(\{0::\text{real}..2\} \cup \{3\} \cup \{5..7\}) = \{0 < .. < 2\} \cup \{5 < .. < 7\}$  (**is** *?lhs = ?rhs*)

**proof** –

**have** *?lhs = interior*  $(\{0..2\} \cup \{5..7\})$

**by** (*metis (no\_types, lifting) Un\_assoc Un\_commute closed\_Un closed\_eucl\_atLeastAtMost interior\_closed\_Un\_empty\_interior\_singleton*)

**also have**  $\dots = ?rhs$

**by** (*simp add: interior\_union\_closed\_intervals*)

**finally show** *?thesis* .

**qed**

The operators can be distinguished by testing which of the points in  $\{1,2,3,4,6\}$  belong to their results.

**definition**

*test* ::  $(\text{real set} \Rightarrow \text{real set}) \Rightarrow \text{bool list}$

**where**

*test* *f*  $\equiv \text{map } (\lambda x. x \in f \text{ RRR}) [1,2,3,4,6]$

**lemma** *RRR\_test*:

**assumes**  $f \text{ RRR} = g \text{ RRR}$

**shows** *test* *f* = *test* *g*

**unfolding** *test\_def* **using** *assms* **by** *simp*

**lemma** *nf\_RRR*:

**shows**

```

test id = [False, False, True, False, True]
test C = [True, True, False, True, False]
test K = [True, True, True, False, True]
test (K ∘ C) = [True, True, True, True, True]
test (C ∘ K) = [False, False, False, True, False]
test (C ∘ K ∘ C) = [False, False, False, False, False]
test (K ∘ C ∘ K) = [False, True, True, True, False]
test (C ∘ K ∘ C ∘ K) = [True, False, False, False, True]
test (K ∘ C ∘ K ∘ C) = [True, True, False, False, False]
test (C ∘ K ∘ C ∘ K ∘ C) = [False, False, True, True, True]
test (K ∘ C ∘ K ∘ C ∘ K) = [True, True, False, False, True]
test (C ∘ K ∘ C ∘ K ∘ C ∘ K) = [False, False, True, True, False]
test (K ∘ C ∘ K ∘ C ∘ K ∘ C) = [False, True, True, True, True]
test (C ∘ K ∘ C ∘ K ∘ C ∘ K ∘ C) = [True, False, False, False, False]

```

**unfolding** *test\_def C\_def K\_def*

**by** (*simp\_all add: RRR\_closure RRR\_interior RRR\_interior\_closure closure\_complement closed\_interval\_Int\_compl o\_apply*)  
(*simp\_all add: RRR\_def*)

**theorem** *CK\_nf\_real\_card*:

**shows** *card ((λ f. f RRR) ‘ {f . CK\_nf f}) = 14*

**by** (*simp add: CK\_nf\_set*) ((*subst card\_insert\_disjoint; auto dest!: RRR\_test simp: nf\_RRR id\_def[symmetric]*)[1])+

**theorem** *CK\_real\_card*:

**shows** *card {f::real set ⇒ real set. CK f} = 14 (is ?lhs = ?rhs)*

**proof**(*rule antisym[OF CK\_card]*)

**show** *?rhs ≤ ?lhs*

**unfolding** *CK\_nf*

**by** (*rule le\_trans[OF eq\_imp\_le[OF CK\_nf\_real\_card[symmetric]] card\_image\_le]*)

(*simp add: CK\_nf\_set*)

**qed**

## 5 A corollary of Kuratowski’s result

We show that it is a corollary of *CK\_real\_card* that at most 7 distinct operators on a topological space can be generated by compositions of closure and interior. In the case of  $\mathbb{R}$ , exactly 7 distinct operators can be so generated.

**inductive** *IK* :: (*'a::topological\_space set ⇒ 'a set*) ⇒ *bool* **where**

```

  IK id
| IK I
| IK K
| [ IK f; IK g ] ⇒ IK (f ∘ g)

```

**inductive** *IK\_nf* :: (*'a::topological\_space set ⇒ 'a set*) ⇒ *bool* **where**

```

  IK_nf id
| IK_nf I
| IK_nf K
| IK_nf (I ∘ K)
| IK_nf (K ∘ I)
| IK_nf (I ∘ K ∘ I)
| IK_nf (K ∘ I ∘ K)

```

**declare** *IK.intros[intro!]*

**declare** *IK\_nf.intros[intro!]*

**lemma** *IK\_nf\_set*:

*{f . IK\_nf f} = {id, I, K, I ∘ K, K ∘ I, I ∘ K ∘ I, K ∘ I ∘ K}*

by (auto simp: *IK\_nf\_simps*)

**theorem** *IK\_nf*:

$IK\ f \longleftrightarrow IK\_nf\ f$

**proof**(rule *iffI*)

assume  $IK\ f$  then show  $IK\_nf\ f$

by induct

(elim *IK\_nf.cases*; clarsimp simp: *id\_def[symmetric]* *o\_assoc*; simp add: *II\_K\_K o\_assoc[symmetric]*; clarsimp simp: *K\_I\_K\_I I\_K\_I\_K o\_assoc*

| *blast*)+

next

assume  $IK\_nf\ f$  then show  $IK\ f$  by induct *blast*+

qed

**theorem** *IK\_card*:

shows  $card\ \{f.\ IK\ f\} \leq 7$

by (auto simp: *IK\_nf IK\_nf\_set card\_insert intro!*: *le\_trans[OF card\_Diff1\_le]*)

**theorem** *IK\_nf\_real\_card*:

shows  $card\ ((\lambda f.\ f\ RRR)\ \{\ f.\ IK\_nf\ f\}) = 7$

by (simp add: *IK\_nf\_set*) ((subst *card\_insert\_disjoint*; auto dest!: *RRR\_test simp: nf\_RRR I\_K id\_def[symmetric] o\_assoc*)[1])+

**theorem** *IK\_real\_card*:

shows  $card\ \{f::real\ set \Rightarrow real\ set.\ IK\ f\} = 7$  (is ?lhs = ?rhs)

**proof**(rule *antisym[OF IK\_card]*)

show ?rhs  $\leq$  ?lhs

unfolding *IK\_nf*

by (rule *le\_trans[OF eq\_refl[OF IK\_nf\_real\_card[symmetric]] card\_image\_le]*)

(simp add: *IK\_nf\_set*)

qed

## 6 Chagrov's result

Chagrov's theorem, which is discussed in Section 2.1 of [Gardner and Jackson \(2008\)](#), states that the number of distinct operators on a topological space that can be generated by compositions of closure and complement is one of 2, 6, 8, 10 or 14.

We begin by observing that the set of normal forms *CK\_nf* can be split into two disjoint sets, *CK\_nf\_pos* and *CK\_nf\_neg*, which we define in terms of interior and closure.

**inductive** *CK\_nf\_pos* :: ('a::topological\_space set  $\Rightarrow$  'a set)  $\Rightarrow$  bool **where**

*CK\_nf\_pos id*  
| *CK\_nf\_pos I*  
| *CK\_nf\_pos K*  
| *CK\_nf\_pos (I  $\circ$  K)*  
| *CK\_nf\_pos (K  $\circ$  I)*  
| *CK\_nf\_pos (I  $\circ$  K  $\circ$  I)*  
| *CK\_nf\_pos (K  $\circ$  I  $\circ$  K)*

**declare** *CK\_nf\_pos.intros*[*intro!*]

**lemma** *CK\_nf\_pos\_set*:

shows  $\{f.\ CK\_nf\_pos\ f\} = \{id,\ I,\ K,\ I \circ K,\ K \circ I,\ I \circ K \circ I,\ K \circ I \circ K\}$

by (auto simp: *CK\_nf\_pos\_simps*)

**definition**

*CK\_nf\_neg* :: ('a::topological\_space set  $\Rightarrow$  'a set)  $\Rightarrow$  bool

**where**



$$CK\_nf\_neg\ f \longleftrightarrow (\exists g. CK\_nf\_pos\ g \wedge f = C \circ g)$$

**lemma** *CK\_nf\_pos\_neg\_disjoint*:

**assumes** *CK\_nf\_pos* *f*

**assumes** *CK\_nf\_neg* *g*

**shows**  $f \neq g$

**using** *assms unfolding CK\_nf\_neg\_def*

**by** (*clarsimp simp: CK\_nf\_pos.simps; elim disjE; metis comp\_def C\_def I\_def K\_def Compl\_iff closure\_UNIV interior\_UNIV id\_apply*)

**lemma** *CK\_nf\_pos\_neg\_CK\_nf*:

$$CK\_nf\ f \longleftrightarrow CK\_nf\_pos\ f \vee CK\_nf\_neg\ f \text{ (is } ?lhs \longleftrightarrow ?rhs)$$

**proof**(*rule iffI*)

**assume** *?lhs* **then show** *?rhs*

**unfolding** *CK\_nf\_neg\_def*

**by** (*rule CK\_nf.cases; metis (no\_types, lifting) CK\_nf\_pos.simps C\_C I\_K K\_I comp\_id o\_assoc*)

**next**

**assume** *?rhs* **then show** *?lhs*

**unfolding** *CK\_nf\_neg\_def*

**by** (*auto elim!: CK\_nf\_pos.cases simp: I\_K C\_C o\_assoc*)

**qed**

We now focus on *CK\_nf\_pos*. In particular, we show that its cardinality for any given topological space is one of 1, 3, 4, 5 or 7.

The proof consists of exhibiting normal forms for the operators supported by each of six classes of topological spaces. These are sublattices of the following lattice of *CK\_nf\_pos* operators:

**lemmas** *K\_I\_K\_subseteq\_K = closure\_mono[OF interior\_subset, of closure X, simplified]* **for** *X*

**lemma** *CK\_nf\_pos\_lattice*:

**shows**

$$I \leq (id :: 'a::topological\_space\ set \Rightarrow 'a\ set)$$

$$id \leq (K :: 'a::topological\_space\ set \Rightarrow 'a\ set)$$

$$I \leq I \circ K \circ (I :: 'a::topological\_space\ set \Rightarrow 'a\ set)$$

$$I \circ K \circ I \leq I \circ (K :: 'a::topological\_space\ set \Rightarrow 'a\ set)$$

$$I \circ K \circ I \leq K \circ (I :: 'a::topological\_space\ set \Rightarrow 'a\ set)$$

$$I \circ K \leq K \circ I \circ (K :: 'a::topological\_space\ set \Rightarrow 'a\ set)$$


$$K \circ I \leq K \circ I \circ (K :: 'a::topological\_space\ set \Rightarrow 'a\ set)$$

$$K \circ I \circ K \leq (K :: 'a::topological\_space\ set \Rightarrow 'a\ set)$$

**unfolding** *I\_def K\_def*

**by** (*simp\_all add: interior\_subset closure\_subset interior\_maximal closure\_mono o\_apply interior\_mono K\_I\_K\_subseteq\_K le\_funI*)

We define the six classes of topological spaces in question, and show that they are related by inclusion in the following way (as shown in Figure 2.3 of Gardner and Jackson (2008)):



## 6.1 Discrete spaces

### definition

$discrete (X :: 'a::topological\_space\ set) \longleftrightarrow I = (id::'a\ set \Rightarrow 'a\ set)$

### lemma *discrete\_eqs*:

**assumes**  $discrete (X :: 'a::topological\_space\ set)$

**shows**

$I = (id::'a\ set \Rightarrow 'a\ set)$

$K = (id::'a\ set \Rightarrow 'a\ set)$

**using** *assms* **unfolding** *discrete\_def* **by** (*auto simp: C-C K-I*)

### lemma *discrete\_card*:

**assumes**  $discrete (X :: 'a::topological\_space\ set)$

**shows**  $card \{f. CK\_nf\_pos (f::'a\ set \Rightarrow 'a\ set)\} = 1$

**using** *discrete\_eqs*[*OF assms*] *CK\_nf\_pos\_lattice*[**where**  $'a='a$ ] **by** (*simp add: CK\_nf\_pos\_set*)

### lemma *discrete\_discrete\_topology*:

**fixes**  $X :: 'a::topological\_space\ set$

**assumes**  $\bigwedge Y::'a\ set. open\ Y$

**shows**  $discrete\ X$

**using** *assms* **unfolding** *discrete\_def I\_def interior\_def islimpt\_def* **by** (*auto simp: fun\_eq\_iff*)

## 6.2 Partition spaces

### definition

$part (X :: 'a::topological\_space\ set) \longleftrightarrow K \circ I = (I :: 'a\ set \Rightarrow 'a\ set)$

### lemma *discrete\_part*:

**assumes**  $discrete\ X$

**shows**  $part\ X$

**using** *assms* **unfolding** *discrete\_def part\_def* **by** (*simp add: C-C K-I*)

### lemma *part\_eqs*:

**assumes**  $part (X :: 'a::topological\_space\ set)$

**shows**

$K \circ I = (I :: 'a\ set \Rightarrow 'a\ set)$

$I \circ K = (K :: 'a\ set \Rightarrow 'a\ set)$

**using** *assms* **unfolding** *part\_def* **by** (*assumption, metis (no\_types, hide\_lams) I-I K-I o\_assoc*)

```

lemma part_not_discrete_card:
  assumes part ( $X :: 'a::\text{topological\_space}$  set)
  assumes  $\neg \text{discrete } X$ 
  shows  $\text{card } \{f. \text{CK\_nf\_pos } (f::'a \text{ set} \Rightarrow 'a \text{ set})\} = 3$ 
using part_eqs[OF  $\langle \text{part } X \rangle$ ]  $\langle \neg \text{discrete } X \rangle$  CK\_nf\_pos\_lattice[where  $'a='a$ ]
unfolding discrete_def
by (simp add: CK\_nf\_pos\_set card_insert_if C_C I_K K_K o\_assoc;metis comp_id)

```

A partition space is a topological space whose basis consists of the empty set and the equivalence classes of points of the space induced by some equivalence relation  $R$  on the underlying set of the space. Equivalently, a partition space is one in which every open set is closed. Thus, for example, the class of partition spaces includes every topological space whose open sets form a boolean algebra.

```

datatype part_witness =  $a \mid b \mid c$ 

```

```

lemma part_witness_UNIV:
  shows  $\text{UNIV} = \text{set } [a, b, c]$ 
using part_witness.exhaust by auto

```

```

lemmas part_witness_pow = subset_subseqs[OF subset_trans[OF subset_UNIV Set.equalityD1[OF part_witness_UNIV]]]

```

```

lemmas part_witness_Compl = Compl_eq_Diff_UNIV[where  $'a=\text{part\_witness}$ , unfolded part_witness_UNIV, simplified]

```

```

instantiation part_witness :: topological_space
begin

```

```

definition open_part_witness  $X \longleftrightarrow X \in \{\{\}, \{a\}, \{b, c\}, \{a, b, c\}\}$ 

```

```

lemma part_witness_ball:
   $(\forall s \in S. s \in \{\{\}, \{a\}, \{b, c\}, \{a, b, c\}\}) \longleftrightarrow S \subseteq \text{set } [\{\}, \{a\}, \{b, c\}, \{a, b, c\}]$ 
by simp blast

```

```

lemmas part_witness_subsets_pow = subset_subseqs[OF iffD1[OF part_witness_ball]]

```

```

instance proof standard
  fix  $K :: \text{part\_witness}$  set set
  assume  $\forall S \in K. \text{open } S$  then show  $\text{open } (\bigcup K)$ 
  unfolding open_part_witness_def
  by  $-$  (drule part_witness_subsets_pow; clarsimp; elim disjE; simp add: insert_commute)
qed (auto simp: open_part_witness_def part_witness_UNIV)

```

```

end

```

```

lemma part_witness_interior_simps:
  shows
    interior  $\{a\} = \{a\}$ 
    interior  $\{b\} = \{\}$ 
    interior  $\{c\} = \{\}$ 
    interior  $\{a, b\} = \{a\}$ 
    interior  $\{a, c\} = \{a\}$ 
    interior  $\{b, c\} = \{b, c\}$ 
    interior  $\{a, b, c\} = \{a, b, c\}$ 
unfolding interior_def open_part_witness_def by auto

```

```

lemma part_witness_part:
  fixes  $X :: \text{part\_witness}$  set
  shows part  $X$ 
proof  $-$ 

```

```

have closure (interior Y) = interior Y for Y :: part_witness set
  using part_witness_pow[where X=Y]
  by (auto simp: closure_interior part_witness_interior_simps part_witness_Compl insert_Diff_if)
then show ?thesis
  unfolding part_def I_def K_def by (simp add: o_def)
qed

```

```

lemma part_witness_not_discrete:
  fixes X :: part_witness set
  shows ¬discrete X
unfolding discrete_def I_def
by (clarsimp simp: o_apply fun_eq_iff exI[where x={b}] part_witness_interior_simps)

```

```

lemma part_witness_card:
  shows card {f. CK_nf_pos (f::part_witness set ⇒ part_witness set)} = 3
by (rule part_not_discrete_card[OF part_witness_part part_witness_not_discrete])

```

### 6.3 Extremally disconnected and open unresolvable spaces

#### definition

```

ed_ou (X :: 'a::topological_space set) ⟷ I ∘ K = K ∘ (I :: 'a set ⇒ 'a set)

```

```

lemma discrete_ed_ou:
  assumes discrete X
  shows ed_ou X
using assms unfolding discrete_def ed_ou_def by simp

```

```

lemma ed_ou_eqs:
  assumes ed_ou (X :: 'a::topological_space set)
  shows
    I ∘ K ∘ I = K ∘ (I :: 'a set ⇒ 'a set)
    K ∘ I ∘ K = K ∘ (I :: 'a set ⇒ 'a set)
    I ∘ K = K ∘ (I :: 'a set ⇒ 'a set)
using assms unfolding ed_ou_def by (metis I I K K o_assoc)+

```

```

lemma ed_ou_neqs:
  assumes ed_ou (X :: 'a::topological_space set)
  assumes ¬discrete X
  shows
    I ≠ (K :: 'a set ⇒ 'a set)
    I ≠ K ∘ (I :: 'a set ⇒ 'a set)
    K ≠ K ∘ (I :: 'a set ⇒ 'a set)
    I ≠ (id :: 'a set ⇒ 'a set)
    K ≠ (id :: 'a set ⇒ 'a set)
using assms CK_nf_pos_lattice[where 'a='a]
unfolding ed_ou_def discrete_def
by (metis (no_types, lifting) C_C I K K I comp_id o_assoc antisym)+

```

```

lemma ed_ou_not_discrete_card:
  assumes ed_ou (X :: 'a::topological_space set)
  assumes ¬discrete X
  shows card {f. CK_nf_pos (f::'a set ⇒ 'a set)} = 4
using ed_ou_eqs[OF ed_ou X] ed_ou_neqs[OF assms]
by (subst CK_nf_pos_set) (subst card_insert_disjoint; (auto)[1])+

```

We consider an example extremally disconnected and open unresolvable topological space.

```

datatype ed_ou_witness = a | b | c | d | e

```

```

lemma ed_ou_witness_UNIV:
  shows  $UNIV = set [a, b, c, d, e]$ 
using ed_ou_witness.exhaust by auto

lemmas ed_ou_witness_pow = subset_subseqs[OF subset_trans[OF subset_UNIV Set.equalityD1 [OF ed_ou_witness_UNIV]]]

lemmas ed_ou_witness_Compl = Compl_eq_Diff_UNIV[where 'a=ed_ou_witness, unfolded ed_ou_witness_UNIV, simplified]

instance ed_ou_witness :: finite
by standard (simp add: ed_ou_witness_UNIV)

instantiation ed_ou_witness :: topological_space
begin

inductive open_ed_ou_witness :: ed_ou_witness set  $\Rightarrow$  bool where
  open_ed_ou_witness {}
| open_ed_ou_witness {a}
| open_ed_ou_witness {b}
| open_ed_ou_witness {c}
| open_ed_ou_witness {a, c}
| open_ed_ou_witness {b, d}
| open_ed_ou_witness {a, c, e}

| open_ed_ou_witness {a, b}
| open_ed_ou_witness {a, e}
| open_ed_ou_witness {b, e}
| open_ed_ou_witness {a, b, c}
| open_ed_ou_witness {a, b, d}
| open_ed_ou_witness {a, b, e}
| open_ed_ou_witness {b, d, e}
| open_ed_ou_witness {a, b, c, d}
| open_ed_ou_witness {a, b, c, e}
| open_ed_ou_witness {a, b, d, e}
| open_ed_ou_witness {a, b, c, d, e}

declare open_ed_ou_witness.intros[intro!]

lemma ed_ou_witness_inter:
  fixes S :: ed_ou_witness set
  assumes open S
  assumes open T
  shows open (S  $\cap$  T)
using assms by (auto elim!: open_ed_ou_witness.cases)

lemma ed_ou_witness_union:
  fixes X :: ed_ou_witness set set
  assumes  $\forall x \in X. open\ x$ 
  shows open ( $\bigcup X$ )
using finite[of X] assms
by (induct, force)
  (clarsimp; elim open_ed_ou_witness.cases; simp add: open_ed_ou_witness.simps subset_insertI2 insert_commute; metis Union_empty_conv)

instance
by standard (auto simp: ed_ou_witness_UNIV intro: ed_ou_witness_inter ed_ou_witness_union)

end

```

**lemma** *ed\_ou\_witness\_interior\_simps*:

**shows**

*interior* {*a*} = {*a*}  
*interior* {*b*} = {*b*}  
*interior* {*c*} = {}  
*interior* {*d*} = {}  
*interior* {*e*} = {*e*}  
*interior* {*a*, *b*} = {*a*, *b*}  
*interior* {*a*, *c*} = {*a*, *c*}  
*interior* {*a*, *d*} = {*a*}  
*interior* {*a*, *e*} = {*a*, *e*}  
*interior* {*b*, *c*} = {*b*}  
*interior* {*b*, *d*} = {*b*, *d*}  
*interior* {*b*, *e*} = {*b*, *e*}  
*interior* {*c*, *d*} = {}  
*interior* {*c*, *e*} = {*e*}  
*interior* {*d*, *e*} = {*e*}  
*interior* {*a*, *b*, *c*} = {*a*, *b*, *c*}  
*interior* {*a*, *b*, *d*} = {*a*, *b*, *d*}  
*interior* {*a*, *b*, *e*} = {*a*, *b*, *e*}  
*interior* {*a*, *c*, *d*} = {*a*, *c*}  
*interior* {*a*, *c*, *e*} = {*a*, *c*, *e*}  
*interior* {*a*, *d*, *e*} = {*a*, *e*}  
*interior* {*b*, *c*, *d*} = {*b*, *d*}  
*interior* {*b*, *c*, *e*} = {*b*, *e*}  
*interior* {*b*, *d*, *e*} = {*b*, *d*, *e*}  
*interior* {*c*, *d*, *e*} = {*e*}  
*interior* {*a*, *b*, *c*, *d*} = {*a*, *b*, *c*, *d*}  
*interior* {*a*, *b*, *c*, *e*} = {*a*, *b*, *c*, *e*}  
*interior* {*a*, *b*, *d*, *e*} = {*a*, *b*, *d*, *e*}  
*interior* {*a*, *b*, *c*, *d*, *e*} = {*a*, *b*, *c*, *d*, *e*}  
*interior* {*a*, *c*, *d*, *e*} = {*a*, *c*, *e*}  
*interior* {*b*, *c*, *d*, *e*} = {*b*, *d*, *e*}

**unfolding** *interior\_def* **by** *safe (clarsimp simp: open\_ed\_ou\_witness\_simps; blast)+*

**lemma** *ed\_ou\_witness\_not\_discrete*:

**fixes** *X* :: *ed\_ou\_witness set*

**shows**  $\neg$ *discrete* *X*

**unfolding** *discrete\_def I\_def* **using** *ed\_ou\_witness\_interior\_simps* **by** (*force simp: fun\_eq\_iff*)

**lemma** *ed\_ou\_witness\_ed\_ou*:

**fixes** *X* :: *ed\_ou\_witness set*

**shows** *ed\_ou* *X*

**unfolding** *ed\_ou\_def I\_def K\_def*

**proof**(*clarsimp simp: o\_apply fun\_eq\_iff*)

**fix** *x* :: *ed\_ou\_witness set*

**from** *ed\_ou\_witness\_pow*[*of x*]

**show** *interior* (*closure x*) = *closure* (*interior x*)

**by**  $-$  (*simp; elim disjE; simp add: closure\_interior ed\_ou\_witness\_interior\_simps ed\_ou\_witness\_Cmpl insert\_Diff\_if*)

**qed**

**lemma** *ed\_ou\_witness\_card*:

**shows** *card* {*f*. *CK\_nf\_pos* (*f*::*ed\_ou\_witness set*  $\Rightarrow$  *ed\_ou\_witness set*)} = 4

**by** (*rule ed\_ou\_not\_discrete\_card[OF ed\_ou\_witness\_ed\_ou ed\_ou\_witness\_not\_discrete]*)

## 6.4 Extremely disconnected spaces

### definition

$extremally\_disconnected (X :: 'a::topological\_space\ set) \iff K \circ I \circ K = I \circ (K :: 'a\ set \Rightarrow 'a\ set)$

### lemma *ed\_ou\_part\_extremally\_disconnected*:

**assumes** *ed\_ou* *X*

**assumes** *part* *X*

**shows** *extremally\_disconnected* *X*

**using** *assms* **unfolding** *extremally\_disconnected\_def ed\_ou\_def part\_def* **by** *simp*

### lemma *extremally\_disconnected\_eqs*:

**fixes** *X* :: *'a::topological\_space set*

**assumes** *extremally\_disconnected* *X*

**shows**

$I \circ K \circ I = K \circ (I :: 'a\ set \Rightarrow 'a\ set)$

$K \circ I \circ K = I \circ (K :: 'a\ set \Rightarrow 'a\ set)$

**using** *assms* **unfolding** *extremally\_disconnected\_def* **by** (*metis* *K\_I\_K\_I*)**+**

### lemma *extremally\_disconnected\_not\_part\_not\_ed\_ou\_card*:

**fixes** *X* :: *'a::topological\_space set*

**assumes** *extremally\_disconnected* *X*

**assumes**  $\neg$ *part* *X*

**assumes**  $\neg$ *ed\_ou* *X*

**shows**  $card \{f. CK\_nf\_pos (f::'a\ set \Rightarrow 'a\ set)\} = 5$

**using** *extremally\_disconnected\_eqs*[*OF*  $\langle$ *extremally\_disconnected* *X* $\rangle$ ] *CK\_nf\_pos\_lattice*[**where** *'a='a*] *assms*(2,3)

**unfolding** *part\_def ed\_ou\_def*

**by** (*simp* *add*: *CK\_nf\_pos\_set C\_C I\_K K\_K o\_assoc card\_insert\_if*; *metis* (*no\_types*) *C\_C K\_I id\_comp o\_assoc*)

Any topological space having an infinite underlying set and whose topology consists of the empty set and every cofinite subset of the underlying set is extremely disconnected. We consider an example such space having a countably infinite underlying set.

**datatype** *'a* *cofinite* = *cofinite 'a*

**instantiation** *cofinite* :: (*type*) *topological\_space*

**begin**

**definition** *open\_cofinite* = ( $\lambda X::'a\ cofinite\ set. finite (-X) \vee X = \{\}$ )

**instance**

**by** *standard* (*auto* *simp*: *open\_cofinite\_def uminus\_Sup*)

**end**

### lemma *cofinite\_closure\_finite*:

**fixes** *X* :: *'a* *cofinite set*

**assumes** *finite* *X*

**shows** *closure* *X* = *X*

**using** *assms* **by** (*simp* *add*: *closed\_open open\_cofinite\_def*)

### lemma *cofinite\_closure\_infinite*:

**fixes** *X* :: *'a* *cofinite set*

**assumes** *infinite* *X*

**shows** *closure* *X* = *UNIV*

**using** *assms* **by** (*metis* *Compl\_empty\_eq closure\_subset double\_compl finite\_subset interior\_complement open\_cofinite\_def open\_interior*)

### lemma *cofinite\_interior\_finite*:

**fixes**  $X :: 'a$  cofinite set  
**assumes** finite  $X$   
**assumes** infinite ( $UNIV :: 'a$  cofinite set)  
**shows** interior  $X = \{\}$   
**using** *assms* cofinite\_closure\_infinite[**where**  $X = -X$ ] **by** (*simp* add: interior\_closure)

**lemma** cofinite\_interior\_infinite:  
**fixes**  $X :: 'a$  cofinite set  
**assumes** infinite  $X$   
**assumes** infinite ( $-X$ )  
**shows** interior  $X = \{\}$   
**using** *assms* cofinite\_closure\_infinite[**where**  $X = -X$ ] **by** (*simp* add: interior\_closure)

**abbreviation** evens :: nat cofinite set  $\equiv \{\text{cofinite } n \mid n. \exists i. n = 2 * i\}$

**lemma** evens\_infinite:  
**shows** infinite evens  
**proof**(rule iffD2[OF infinite\_iff\_countable\_subset], rule exI, rule conjI)  
**let**  $?f = \lambda n :: \text{nat}. \text{cofinite } (2 * n)$   
**show** inj  $?f$  **by** (auto intro: inj\_onI)  
**show** range  $?f \subseteq \text{evens}$  **by** auto  
**qed**

**lemma** cofinite\_nat\_infinite:  
**shows** infinite ( $UNIV :: \text{nat}$  cofinite set)  
**using** evens\_infinite finite\_Diff2 **by** fastforce

**lemma** evens\_Coimpl\_infinite:  
**shows** infinite ( $- \text{evens}$ )  
**proof**(rule iffD2[OF infinite\_iff\_countable\_subset], rule exI, rule conjI)  
**let**  $?f = \lambda n :: \text{nat}. \text{cofinite } (2 * n + 1)$   
**show** inj  $?f$  **by** (auto intro: inj\_onI)  
**show** range  $?f \subseteq -\text{evens}$  **by** clarsimp presburger  
**qed**

**lemma** evens\_closure:  
**shows** closure evens =  $UNIV$   
**using** evens\_infinite **by** (rule cofinite\_closure\_infinite)

**lemma** evens\_interior:  
**shows** interior evens =  $\{\}$   
**using** evens\_infinite evens\_Coimpl\_infinite **by** (rule cofinite\_interior\_infinite)

**lemma** cofinite\_not\_part:  
**fixes**  $X :: \text{nat}$  cofinite set  
**shows**  $\neg \text{part } X$   
**unfolding** part\_def I\_def K\_def  
**using** cofinite\_nat\_infinite  
**by** (clarsimp simp: fun\_eq\_iff o\_apply)  
 (metis (no\_types) cofinite\_closure\_finite cofinite\_interior\_finite double\_coimpl finite.emptyI finite.insertI insert\_not\_empty interior\_closure)

**lemma** cofinite\_not\_ed\_ou:  
**fixes**  $X :: \text{nat}$  cofinite set  
**shows**  $\neg \text{ed\_ou } X$   
**unfolding** ed\_ou\_def I\_def K\_def  
**by** (clarsimp simp: fun\_eq\_iff o\_apply evens\_closure evens\_interior exI[**where**  $x = \text{evens}$ ])



**lemma** *cofinite\_extremally\_disconnected\_aux*:

**fixes**  $X :: \text{nat cofinite set}$

**shows**  $\text{closure} (\text{interior} (\text{closure } X)) \subseteq \text{interior} (\text{closure } X)$

**by** (*metis subsetI closure\_closure closure\_complement closure\_def closure\_empty finite\_Un interior\_eq open\_cofinite\_def open\_interior*)

**lemma** *cofinite\_extremally\_disconnected*:

**fixes**  $X :: \text{nat cofinite set}$

**shows** *extremally\_disconnected*  $X$

**unfolding** *extremally\_disconnected\_def I\_def K\_def*

**by** (*auto simp: fun\_eq\_iff o\_apply dest: subsetD[OF closure\_subset] subsetD[OF interior\_subset] subsetD[OF cofinite\_extremally\_disconnected\_aux]*)

**lemma** *cofinite\_card*:

**shows**  $\text{card} \{f. \text{CK\_nf\_pos} (f :: \text{nat cofinite set} \Rightarrow \text{nat cofinite set})\} = 5$

**by** (*rule extremally\_disconnected\_not\_part\_not\_ed\_ou\_card[OF cofinite\_extremally\_disconnected cofinite\_not\_part cofinite\_not\_ed\_ou]*)

## 6.5 Open unresolvable spaces

**definition**

*open\_unresolvable* ( $X :: 'a::\text{topological\_space set}$ )  $\longleftrightarrow K \circ I \circ K = K \circ (I :: 'a \text{ set} \Rightarrow 'a \text{ set})$

**lemma** *ed\_ou\_open\_unresolvable*:

**assumes** *ed\_ou*  $X$

**shows** *open\_unresolvable*  $X$

**using** *assms unfolding open\_unresolvable\_def* **by** (*simp add: ed\_ou\_eqs*)

**lemma** *open\_unresolvable\_eqs*:

**assumes** *open\_unresolvable* ( $X :: 'a::\text{topological\_space set}$ )

**shows**

$I \circ K \circ I = I \circ (K :: 'a \text{ set} \Rightarrow 'a \text{ set})$

$K \circ I \circ K = K \circ (I :: 'a \text{ set} \Rightarrow 'a \text{ set})$

**using** *assms unfolding open\_unresolvable\_def* **by**  $-$  (*metis I\_K\_I\_K o\_assoc; simp*)

**lemma** *not\_ed\_ou\_neqs*:

**assumes**  $\neg \text{ed\_ou}$  ( $X :: 'a::\text{topological\_space set}$ )

**shows**

$I \neq I \circ (K :: 'a \text{ set} \Rightarrow 'a \text{ set})$

$K \neq K \circ (I :: 'a \text{ set} \Rightarrow 'a \text{ set})$

**using** *assms unfolding ed\_ou\_def*

**by** (*simp\_all add: fun\_eq\_iff I\_K K\_def C\_def o\_apply*)

(*metis (no\_types, hide\_lams) closure\_eq\_empty disjoint\_eq\_subset\_Compl double\_complement interior\_Int interior\_complement set\_eq\_subset*) $+$

**lemma** *open\_unresolvable\_not\_ed\_ou\_card*:

**assumes** *open\_unresolvable* ( $X :: 'a::\text{topological\_space set}$ )

**assumes**  $\neg \text{ed\_ou}$   $X$

**shows**  $\text{card} \{f. \text{CK\_nf\_pos} (f :: 'a \text{ set} \Rightarrow 'a \text{ set})\} = 5$

**using** *open\_unresolvable\_eqs[OF <open\_unresolvable X>] not\_ed\_ou\_neqs[OF <\neg ed\_ou X>] <\neg ed\_ou X>*

**unfolding** *ed\_ou\_def* **by** (*auto simp: CK\_nf\_pos\_set card\_insert\_if*)

We show that the class of open unresolvable spaces is non-empty by exhibiting an example of such a space.

**datatype** *ou\_witness* =  $a \mid b \mid c$

**lemma** *ou\_witness\_UNIV*:

**shows**  $\text{UNIV} = \text{set } [a, b, c]$

**using** *ou\_witness.exhaust* **by** *auto*

**instantiation** *ou\_witness* :: *topological\_space*  
**begin**

**definition** *open\_ou\_witness*  $X \longleftrightarrow a \notin X \vee X = UNIV$

**instance**  
**by** *standard* (*auto simp: open\_ou\_witness\_def*)

**end**

**lemma** *ou\_witness\_closure\_simps*:

**shows**

*closure* {*a*} = {*a*}  
*closure* {*b*} = {*a*, *b*}  
*closure* {*c*} = {*a*, *c*}  
*closure* {*a*, *b*} = {*a*, *b*}  
*closure* {*a*, *c*} = {*a*, *c*}  
*closure* {*a*, *b*, *c*} = {*a*, *b*, *c*}  
*closure* {*b*, *c*} = {*a*, *b*, *c*}

**unfolding** *closure\_def islimpt\_def open\_ou\_witness\_def* **by** *force+*

**lemma** *ou\_witness\_open\_unresolvable*:

**fixes** *X* :: *ou\_witness set*

**shows** *open\_unresolvable X*

**unfolding** *open\_unresolvable\_def I\_def K\_def*

**by** (*clarsimp simp: o\_apply fun\_eq\_iff*)

(*metis (no\_types, lifting) Compl\_iff K\_I\_K\_subseteq\_K closure\_complement closure\_interior closure\_mono closure\_subset interior\_eq interior\_maximal open\_ou\_witness\_def subset\_antisym*)

**lemma** *ou\_witness\_not\_ed\_ou*:

**fixes** *X* :: *ou\_witness set*

**shows**  $\neg ed\_ou\ X$

**unfolding** *ed\_ou\_def I\_def K\_def*

**by** (*clarsimp simp: o\_apply fun\_eq\_iff*)

(*metis UNIV\_I insert\_iff interior\_eq open\_ou\_witness\_def singletonD*  
*ou\_witness.distinct(4,5) ou\_witness.simps(2) ou\_witness\_closure\_simps(2)*)

**lemma** *ou\_witness\_card*:

**shows**  $card\ \{f.\ CK\_nf\_pos\ (f::ou\_witness\ set \Rightarrow ou\_witness\ set)\} = 5$

**by** (*rule open\_unresolvable\_not\_ed\_ou\_card[OF ou\_witness\_open\_unresolvable ou\_witness\_not\_ed\_ou]*)

## 6.6 Kuratowski spaces

**definition**

*kuratowski* ( $X :: 'a::topological\_space\ set$ )  $\longleftrightarrow$   
 $\neg extremally\_disconnected\ X \wedge \neg open\_unresolvable\ X$

A Kuratowski space distinguishes all 7 positive operators.

**lemma** *part\_closed\_open*:

**fixes** *X* ::  $'a::topological\_space\ set$

**assumes**  $I \circ K \circ I = (I::'a\ set \Rightarrow 'a\ set)$

**assumes** *closed X*

**shows** *open X*

**proof**(*rule Topological\_Spaces.openI*)

**fix** *x* **assume**  $x \in X$

**let**  $?S = I\ (\neg\{x\})$

**let**  $?G = -K\ ?S$

**have**  $x \in ?G$   
**proof** –  
**from**  $\langle I \circ K \circ I = I \rangle$  **have**  $I (K (I ?S)) = ?S I ?S = ?S$   
**unfolding**  $I\_def K\_def$  **by**  $(simp\_all \text{ add: } o\_def \text{ fun\_eq\_iff})$   
**then have**  $K (I ?S) \neq UNIV$   
**unfolding**  $I\_def K\_def$  **using**  $interior\_subset$  **by**  $fastforce$   
**moreover have**  $G \subseteq ?S \vee x \in G$  **if open**  $G$  **for**  $G$   
**using that unfolding**  $I\_def$  **by**  $(meson \text{ interior\_maximal } subset\_Compl\_singleton)$   
**ultimately show**  $?thesis$   
**unfolding**  $I\_def K\_def$   
**by**  $clarsimp (metis (no\_types, \text{lifting}) ComplD Compl\_empty\_eq \text{ closure\_interior } \text{ closure\_subset } \text{ ex\_in\_conv } \text{ open\_interior } \text{ subset\_eq})$   
**qed**  
**moreover from**  $\langle I \circ K \circ I = I \rangle$  **have**  $open ?G$   
**unfolding**  $I\_def K\_def$  **by**  $(auto \text{ simp: } fun\_eq\_iff \text{ o\_apply})$   
**moreover have**  $?G \subseteq X$   
**proof** –  
**have**  $?G \subseteq K ?G$  **unfolding**  $K\_def$  **using**  $closure\_subset$  **by**  $fastforce$   
**also from**  $\langle I \circ K \circ I = I \rangle$  **have**  $\dots = K \{x\}$   
**unfolding**  $I\_def K\_def$  **by**  $(metis \text{ closure\_interior } \text{ comp\_def } \text{ double\_complement})$   
**also from**  $\langle closed X \rangle \langle x \in X \rangle$  **have**  $\dots \subseteq X$   
**unfolding**  $K\_def$  **by**  $clarsimp (meson \text{ closure\_minimal } \text{ contra\_subsetD } \text{ empty\_subsetI } \text{ insert\_subset})$   
**finally show**  $?thesis$  .  
**qed**  
**ultimately show**  $\exists T. \text{ open } T \wedge x \in T \wedge T \subseteq X$  **by**  $blast$   
**qed**

**lemma**  $part\_I\_K\_I$ :

**assumes**  $I \circ K \circ I = (I :: 'a :: topological\_space \text{ set} \Rightarrow 'a \text{ set})$   
**shows**  $I \circ K = (K :: 'a \text{ set} \Rightarrow 'a \text{ set})$   
**using**  $interior\_open[OF \text{ part\_closed\_open}[OF \text{ assms } \text{ closed\_closure}]$  **unfolding**  $I\_def K\_def o\_def$  **by**  $simp$

**lemma**  $part\_K\_I\_I$ :

**assumes**  $I \circ K \circ I = (I :: 'a :: topological\_space \text{ set} \Rightarrow 'a \text{ set})$   
**shows**  $K \circ I = (I :: 'a \text{ set} \Rightarrow 'a \text{ set})$   
**using**  $part\_I\_K\_I[OF \text{ assms}] \text{ assms}$  **by**  $simp$

**lemma**  $kuratowski\_neqs$ :

**assumes**  $kuratowski (X :: 'a :: topological\_space \text{ set})$   
**shows**  
 $I \neq I \circ K \circ (I :: 'a \text{ set} \Rightarrow 'a \text{ set})$   
 $I \circ K \circ I \neq K \circ (I :: 'a \text{ set} \Rightarrow 'a \text{ set})$   
 $I \circ K \circ I \neq I \circ (K :: 'a \text{ set} \Rightarrow 'a \text{ set})$   
 $I \circ K \neq K \circ I \circ (K :: 'a \text{ set} \Rightarrow 'a \text{ set})$   
 $K \circ I \neq K \circ I \circ (K :: 'a \text{ set} \Rightarrow 'a \text{ set})$   
 $K \circ I \circ K \neq (K :: 'a \text{ set} \Rightarrow 'a \text{ set})$   
 $I \circ K \neq K \circ (I :: 'a \text{ set} \Rightarrow 'a \text{ set})$   
 $I \neq (id :: 'a \text{ set} \Rightarrow 'a \text{ set})$   
 $K \neq (id :: 'a \text{ set} \Rightarrow 'a \text{ set})$   
 $I \circ K \circ I \neq (id :: 'a \text{ set} \Rightarrow 'a \text{ set})$   
 $K \circ I \circ K \neq (id :: 'a \text{ set} \Rightarrow 'a \text{ set})$   
**using**  $assms \text{ unfolding } \text{kuratowski\_def } \text{extremally\_disconnected\_def } \text{open\_unresolvable\_def}$   
**by**  $(metis (no\_types, \text{lifting}) I\_K K\_K I\_K\_I\_K K\_I\_K\_I \text{ part\_I\_K\_I } \text{ part\_K\_I\_I } \text{ o\_assoc } \text{ comp\_id})+$

**lemma**  $kuratowski\_card$ :

**assumes**  $kuratowski (X :: 'a :: topological\_space \text{ set})$   
**shows**  $card \{f. CK\_nf\_pos (f :: 'a \text{ set} \Rightarrow 'a \text{ set})\} = 7$   
**using**  $CK\_nf\_pos\_lattice[where 'a='a] \text{kuratowski\_neqs}[OF \text{ assms}] \text{ assms}$

**unfolding** *kuratowski\_def extremally\_disconnected\_def open\_unresolvable\_def*  
**by** (*subst CK\_nf\_pos\_set*) (*subst card\_insert\_disjoint*; (*auto*)[1])+

$\mathbb{R}$  is a Kuratowski space.

**lemma** *kuratowski\_reals*:

**shows** *kuratowski* ( $\mathbb{R} :: \text{real set}$ )

**unfolding** *kuratowski\_def extremally\_disconnected\_def open\_unresolvable\_def*

**by** (*rule conjI*)

(*metis* (*no\_types*, *lifting*) *LK list.inject nf\_RRR(11) nf\_RRR(8) o\_assoc*,

*metis* (*no\_types*, *lifting*) *LK fun.map\_comp list.inject nf\_RRR(11) nf\_RRR(9)*)

## 6.7 Chagrov's theorem

**theorem** *chagrov*:

**fixes**  $X :: 'a::\text{topological\_space set}$

**obtains** *discrete X*

|  $\neg \text{discrete } X \wedge \text{part } X$

|  $\neg \text{discrete } X \wedge \text{ed\_ou } X$

|  $\neg \text{ed\_ou } X \wedge \text{open\_unresolvable } X$

|  $\neg \text{ed\_ou } X \wedge \neg \text{part } X \wedge \text{extremally\_disconnected } X$

| *kuratowski X*

**unfolding** *kuratowski\_def* **by** *metis*

**corollary** *chagrov\_card*:

**shows**  $\text{card } \{f. \text{CK\_nf\_pos } (f::'a::\text{topological\_space set}) \Rightarrow 'a \text{ set}\} \in \{1,3,4,5,7\}$

**using** *discrete\_card part\_not\_discrete\_card ed\_ou\_not\_discrete\_card open\_unresolvable\_not\_ed\_ou\_card*

*extremally\_disconnected\_not\_part\_not\_ed\_ou\_card kuratowski\_card*

**by** (*cases rule: chagrov*) *blast+*

## References

- L. K. Bagińska and A. Grabowski. On the Kuratowski closure-complement problem. *Journal of Formalized Mathematics*, 15, 2003. URL <http://mizar.org/JFM/Vol15/kurato.1.html>. MML Identifier: KURATO\_1.
- A. V. Chagrov. Kuratowski numbers. *Application of functional analysis in approximation theory*, page 186–190, 1982. Gos. Univ., Kalinin [Russian].
- M. Chamberland. *Single Digits: In Praise of Small Numbers*. Princeton University Press, 2015.
- B.J. Gardner and M. Jackson. The Kuratowski closure-complement theorem. *New Zealand Journal of Mathematics*, 38:9–44, 2008. URL [http://nzjm.math.auckland.ac.nz/images/6/63/The\\_Kuratowski\\_Closure-Complement\\_Theorem.pdf](http://nzjm.math.auckland.ac.nz/images/6/63/The_Kuratowski_Closure-Complement_Theorem.pdf).
- A. Grabowski. Solving two problems in general topology via types. In J.-C. Filliâtre, C. Paulin-Mohring, and B. Werner, editors, *TYPES 2004*, volume 3839 of *LNCS*, pages 138–153. Springer, 2004.
- K. Kuratowski. Sur l'opération  $\bar{A}$  de l'analysis situs. *Fundamenta Mathematicae*, (3):182–199, 1922.
- D. Rusin, June 2001. URL <http://web.archive.org/web/20031011151110/http://www.math.niu.edu/~rusin/known-math/94/kuratowski>.
- R. Whitty. Kuratowski's 14-set theorem, 2015. URL <http://www.theoremoftheday.org/Topology/Kuratowski14/TotDKuratowski14.pdf>.